*** Generating figure 3: Baseline results for the full sample ***

** Preamble **

clear all
set more off


local path_input  "C:/Users/$user/Dropbox/$project/master/output/data"
local path_output "C:/Users/$user/Dropbox/$project/master/output/tables"

* Note: local directory is the output folder 
cd "`path_output'"

					** Loading the input file **


use `path_input'/dataset_final_may20, clear

estimates clear

xtset codigo_ibge turno

local baseline_controls "pop1991 years_sch1991 renda1991 sh_tv1991 sh_agua1991 sh_elect1991 sh_rural"
local CGB_FEs			"pct_sh_tv1991 pct_sh_rural pct_renda1991"


/*generating percentiles*/
foreach var in `baseline_controls' {
xtile pct_`var'=  `var', n(10)
replace pct_`var'=0 if turno==2
}

/*generating pscores*/
encode uf, gen(UF)
psmatch2 Globo T1* if turno==1,  noreplace neighbor(1)
egen ppp=mean(_pscore), by(codigo)
gen pscore=1
replace pscore=ppp/(1-ppp) if Globo==0

label var treatment "Globo"
label var treatment1 "Globo and Others"
label var treatment2 "Globo Only"

foreach v of varlist * {

	label variable `v' `"\hspace{0.1cm} `: variable label `v''"'
	
}

										** Table A1 **
										
/*
The table reports regression estimates of the effect of Globo's edited coverage on the second round's vote shares in the 1989 presidential elections. All specification controls for municipality and time fixed effects. In panel A, columns (1) and (2) reports baseline estimates using Lula's vote share as an outcome, and (3) and (4) using Collor's vote share. Specifications in panel A control for first-round vote-shares interacted with the time fixed effects. In panel B, columns (1) and (2) reports baseline estimates using left-wing candidates' vote share as an outcome, and (3) and (4) using right-wing candidates' vote shares. Columns (2) and (4) splits treatment status according to coverage of other TV broadcasters. Heteroskedasticity-adjusted standard errors clustered at the municipality level are reported in brackets. Significantly different from zero at $99\% (***)$, $95\% (**)$, and $90\% (*)$ confidence level.
*/

local legend "The table reports regression estimates of the effect of Globo's edited coverage on the second round's vote shares in the 1989 presidential elections. All specification controls for municipality and time fixed effects. In panel A, columns (1) and (2) reports baseline estimates using Lula's vote share as an outcome, and (3) and (4) using Collor's vote share. Specifications in panel A control for first-round vote-shares interacted with the time fixed effects. In panel B, columns (1) and (2) reports baseline estimates using left-wing candidates' vote share as an outcome, and (3) and (4) using right-wing candidates' vote shares. Columns (2) and (4) splits treatment status according to coverage of other TV broadcasters. Heteroskedasticity-adjusted standard errors clustered at the municipality level are reported in brackets. Significantly different from zero at $99\% (***)$, $95\% (**)$, and $90\% (*)$ confidence level."										

									
										** Panel A **
						
eststo: xi: reghdfe sh_Tpt_  treatment 			   sh_Tpsp_-sh_Tpsdb_ turno, vce(cluster codigo_ibge) a(codigo_ibge)
estadd local Pvalue "-"
estadd local PartiesControl "Yes"

eststo: xi: reghdfe sh_Tpt_  treatment1 treatment2 sh_Tpsp_-sh_Tpsdb_ turno, vce(cluster codigo_ibge) a(codigo_ibge)
test treatment1 - treatment2 = 0
scalar Pvalue = round(r(p), 0.001)
estadd scalar Pvalue
estadd local PartiesControl "Yes"

eststo: xi: reghdfe sh_Tprn_ treatment 		       sh_Tpsp_-sh_Tpsdb_ turno, vce(cluster codigo_ibge) a(codigo_ibge)
estadd local Pvalue "-"
estadd local PartiesControl "Yes"

eststo: xi: reghdfe sh_Tprn_ treatment1 treatment2 sh_Tpsp_-sh_Tpsdb_ turno, vce(cluster codigo_ibge) a(codigo_ibge)
test treatment1 - treatment2 = 0
scalar Pvalue = round(r(p), 0.001)
estadd scalar Pvalue
estadd local PartiesControl "Yes"



										** Panel B **
									
eststo: xi: reghdfe sh_Tesquerda treatment 			   turno, vce(cluster codigo_ibge) a(codigo_ibge)
estadd local Pvalue "-"
estadd local PartiesControl "No"

eststo: xi: reghdfe sh_Tesquerda treatment1 treatment2 turno, vce(cluster codigo_ibge) a(codigo_ibge)
test treatment1 - treatment2 = 0
scalar Pvalue = round(r(p), 0.001)
estadd scalar Pvalue
estadd local PartiesControl "No"

eststo: xi: reghdfe sh_Tdireita  treatment 			   turno, vce(cluster codigo_ibge) a(codigo_ibge)
estadd local Pvalue "-"
estadd local PartiesControl "No"

eststo: xi: reghdfe sh_Tdireita  treatment1 treatment2 turno, vce(cluster codigo_ibge) a(codigo_ibge)
test treatment1 - treatment2 = 0
scalar Pvalue = round(r(p), 0.001)
estadd scalar Pvalue
estadd local PartiesControl "No"					


esttab est1 est2 est3 est4 using Table_baseline_no_weights_no_fes_panel_A.tex, ///
keep(treatment treatment1 treatment2) ///
order(treatment treatment1 treatment2) ///
stat(N r2 Pvalue PartiesControl, fmt(0 2 2 0) layout("\multicolumn{1}{c}{@}" "\multicolumn{1}{S}{@}" "\multicolumn{1}{S}{@}" "\multicolumn{1}{c}{@}") ///
label("Observations" "R-squared" "Coeff. equality (\emph{p-value})" "Electoral controls * Round")) ///
mgroups("Lula's vote-share" "Collor's vote-share" , pattern(1 0 1 0) span prefix(\multicolumn{@span}{c}{) suffix(}) erepeat(\cline{@span})) ///
star(* 0.10 ** 0.05 *** 0.01) ///
nomtitles  ///
r2 b(3) replace  se(3) brac compress nonotes label  nogaps staraux ///
nonotes addnotes("\begin{minipage}{.8\linewidth} \footnotesize \smallskip \textbf{Note:} `legend' \end{minipage}" )


esttab est5 est6 est7 est8 using Table_baseline_no_weights_no_fes_panel_B.tex, ///
keep(treatment treatment1 treatment2) ///
order(treatment treatment1 treatment2) ///
stat(N r2 Pvalue PartiesControl, fmt(0 2 2 0) layout("\multicolumn{1}{c}{@}" "\multicolumn{1}{S}{@}" "\multicolumn{1}{S}{@}" "\multicolumn{1}{c}{@}") ///
label("Observations" "R-squared" "Coeff. equality (\emph{p-value})" "Electoral controls * Round")) ///
mgroups("Left's vote-share" "Right's vote-share" , pattern(1 0 1 0) span prefix(\multicolumn{@span}{c}{) suffix(}) erepeat(\cline{@span})) ///
star(* 0.10 ** 0.05 *** 0.01) ///
nomtitles  ///
r2 b(3) replace  se(3) brac compress nonotes label  nogaps staraux ///
nonotes addnotes("\begin{minipage}{.8\linewidth} \footnotesize \smallskip \textbf{Note:} `legend' \end{minipage}" )

include "https://raw.githubusercontent.com/steveofconnell/PanelCombine/master/PanelCombine.do"

panelcombine, use(Table_baseline_no_weights_no_fes_panel_A.tex Table_baseline_no_weights_no_fes_panel_B.tex) ///
paneltitles("Lula vs Collor" "Left-wing vs Right-wing") columncount(5) save(Table_Without_Pscore_Conditional.tex) cleanup

